Lab: Building the Contact Management Application

In this lab, you will create a multi-page Web application that uses a database to store contact and call information. The application lets you add contacts, view and add calls made by a contact, and add new contact types. This application uses the Contacts database (Contacts.mdb) that is included with this book' s CD-ROM.

The Contact Management Application demonstrates these key concepts:

· Navigation.  Users navigate from a central switchboard page to specific task pages. Once users are done with a task, they navigate back to the switchboard page.

· Central data connection.  The database connection, data adapter, and data set components reside in the Global.asax file and are initialized once per user. These objects are shared between Web forms using session state variables.

· Data set transactions.  All changes to the database are processed through data sets, and all updates are contained within error-handling structures. If an update fails, the user can try again or cancel the operation without damaging the integrity of the database.

· Bound and unbound controls.  Calls are displayed using a DataList control containing data-bound label controls. Contacts are displayed in a drop-down list control using a code rather than data binding.

· Database commands.  The application gets unique key values by executing commands on the database.

Estimated lesson time: 90 minutes

Exercise 1: Start a Project and Create the SwitchBoard Form

In this exercise, you start a new project and create the Web form used to navigate to the other Web forms in the application. This form is very simple and does not include any executable code.

When you are done creating the SwitchBoard form, it will appear as shown in Figure 5-23.

Figure 5-23 The SwitchBoard form

To start the new application and create the SwitchBoard Web form:

1. Create a new ASP.NET Web application project named ContactManagement.

2. Change the name of the Webform1.aspx file to SwitchBoard.aspx and display the Web form in HTML mode.

3. Type the following HTML between the <form> and </form> tags.

4. <h2>Contact Management Switchboard</h2>

5.    <P>What would you like to do? </P>

6.    <P><a href="Calls.aspx">View/add calls.</a></P>

7.    <P><a href="AddContact.aspx">Add a contact.</a></P>

8.    <P><a href="DeleteContact.aspx">Delete a contact.</a></P>

   <P><a href="ContactTypes.aspx">Add a contact type.</a></P>

Exercise 2: Adding Database Components

The Contact Management application performs data access from several different Web forms. The database connection, adapter, and data set components are added to the Global.asax file and saved as session state variables so they can be accessed from the respective Web forms.

All the Web forms, and all users of the Web forms, share a single database connection. This conserves resources and makes it easier to maintain connection and adapter settings while developing the application because the database components are all located in one place: the Global.asax file.

To add the database components for the Contact Management application

1. Open the Global.asax file and click the Server Explorer link. Visual Studio displays the Server Explorer.

2. Click the Connect To A Database button on the Server Explorer window. Visual Studio displays the DataLink Properties dialog box.

3. Select the name of the SQL server where the Contacts sample database is installed, choose the Use Windows NT integrated security option, and then type Contacts in the Select database on the Server text box. Click Test Connection to verify your settings, then click OK when you are done. Visual Studio adds the database connection to the Server Explorer.

4. In the Server Explorer window, expand the items under the data connection you created by clicking on the plus signs to the left of the Contacts.dbo data connection and then do the same to the Tables sub item. Drag the Calls, ContactTypes, and Contacts tables from the Solution Explorer to the Global.asax page. Visual Studio adds a database connection component and data adapter components for each of the tables.

5. Each data adapter represents one of the tables you dragged onto the page. Right-click one of the data adapters and select Generate Data Set from the pop-up menu. Visual Studio displays the Generate DataSet dialog box.

6. Type the name of the data set to create in the New text box. Name the data set after the table it represents (for example, dsCalls for the Calls table), then select the Add This Dataset To The Designer check box and click OK. Visual Studio adds the data set to the Global.asax page.

	When Visual Studio adds a data set, it appends a number to the name you provide. You need to change the name of the data set in the properties window to remove the appended number.

7. Repeat steps 5 and 6 for each of the data adapters.

8. Double-click the Globabl.asax page in the Code window to display the code for the page. Add the following Session_Start event procedure to create the session state variables that you will use from subsequent Web forms to access the database components:

Visual Basic .NET

Sub Session_Start(ByVal sender As Object, ByVal e As EventArgs)

    ' Database connection.

    Session("ContactMgmt") = SqlConnection1

    ' Data adapters.

    Session("adptCalls") = SqlDataAdapter1

    Session("adptContactTypes") = SqlDataAdapter2

    Session("adptContacts") = SqlDataAdapter3

    ' Data sets.

    Session("dsCalls") = dsCalls

    Session("dsContactTypes") = dsContactTypes

    Session("dsContacts") = dsContacts

End Sub

Visual C#

protected void Session_Start(Object sender, EventArgs e)

{

     // Database connection.

     Session["ContactMgmt"] = SqlConnection1;

     // Data adapters.

     Session["adptCalls"] = SqlDataAdapter1;

     Session["adptContactTypes"] = SqlDataAdapter2;

     Session["adptContacts"] = SqlDataAdapter3;

     // Data sets.

     Session["dsCalls"] = dsCalls;

     Session["dsContactTypes"] = dsContactTypes;

     Session["dsContacts"] = dsContacts;

}

Exercise 3: Create the AddContact Form

The Contact Management application allows users to add information about new contacts through the AddContact Web form. The AddContact Web form is a straightforward data entry form with text boxes for each of the data items in a Contact table record. When complete, the AddContact Web form will appear as shown in Figure 5-24.

[image: image7.png][ Fie et vew Fovotes Toos Heb

Q.0 ¥ @ B

J Back  Fond  Stop  Refresh  Home

L % @ O

Search Favortes Meda Hstary

B.

wal

=
s
Pt Edt  Discuss

| acress [ ] utp:focaios fcscwebappsichapterdsvbContsctitanagement/AddCantact aspx

EBE

Add a Contact

First name:

Last name:

Address =

State: [AL 7]

City.

P

Home phone.
Workphone[
Notes =
Contact type: [Business =]
& Cancel

=< Return to Switchboard

Eloone

[ Jtocalmtranet





Figure 5-24 The AddContact Web form

To create the AddContact data entry Web form

1. Add a new Web form to the project named AddContact.aspx.

2. In the Properties window, select the DOCUMENT object and set the pageLayout property to FlowLayout.

3. Add controls to the Web form with the property settings shown in the following table: 

	Control
	Property
	Setting

	TextBox
	ID
	txtFirstName

	TextBox
	ID
	txtLastName

	TextBox
	ID
	txtAddress

	
	TextMode
	MultiLine

	TextBox
	ID
	txtCity

	DropDownList
	ID
	drpStates

	
	DataSource
	arrState

	TextBox
	ID
	txtZIP

	TextBox
	ID
	txtHomePhone

	TextBox
	ID
	txtWorkPhone

	TextBox
	ID
	txtNotes

	DropDownList
	ID
	drpContactTypes

	Button
	ID
	butAdd

	
	Text
	Add

	Button
	ID
	butCancel

	
	Text
	Cancel

	Literal
	ID
	litStatus

	Hyperlink
	Text
	Return to SwitchBoard

	
	NavigateUrl
	SwitchBoard.aspx


4. Type text to identify the fields directly on the Web form. Use carriage returns to start new lines.

5. Double-click the Web form to display the Code window, and then add the following Imports or using statement to the top of the module:

Visual Basic .NET

Imports System.Data.SqlClient

Visual C#

using System.Data.SqlClient;

6. The AddContact Web form uses the database components created in Exercise 2 on the Global.asax file. To use these components in the Web form, declare class-level variables, and then retrieve the references to the components from the state variables where they are stored. The following code demonstrates how to do this:

Visual Basic .NET

Dim ContactMgmt As SqlConnection

Dim adptContactTypes As SqlDataAdapter

Dim adptContacts As SqlDataAdapter

Dim dsContacts As New dsContacts()

' These variables are public for data binding.

Public dsContactTypes As New dsContactTypes()

Public arrState As String() = {"AL", "AK", "AR", "AZ", _

  "CA", "CO", "CT", "DE", "FL", "GA", "HI", "ID", "IL", _

  "IN", "IA", "KS", "KY", "LA", "MA", "ME", "MD", "MI", _

  "MN", "MO", "MS", "MT", "NC", "ND", "NE", "NH", "NJ", _

  "NM", "NV", "NY", "OH", "OK", "OR", "PA", "RI", "SC", _

  "SD", "TN", "TX", "UT", "VA", "VT", "WA", "WI", "WY"}

Private Sub Page_Load(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles MyBase.Load

    ' Get the connection and adapters from Session state.

    ContactMgmt = Session("ContactMgmt")

    adptContactTypes = Session("adptContactTypes")

    adptContacts = Session("adptContacts")

    ' Fill the ContactType data set.

    adptContactTypes.Fill(dsContactTypes)

    ' Fill the Contacts data set.

    adptContacts.Fill(dsContacts)

    ' Bind to data -- populates the drpContactTypes and drpState lists.

    drpContactTypes.DataBind()

    drpStates.DataBind()

End Sub

Visual C#

SqlConnection ContactMgmt;

SqlDataAdapter adptContactTypes;

SqlDataAdapter adptContacts;

dsContacts dsContacts = new dsContacts();

// These variables are public for data binding.

public dsContactTypes dsContactTypes = new dsContactTypes();

public string[] arrState = {"AL", "AK", "AR", "AZ", 

  "CA", "CO", "CT", "DE", "FL", "GA", "HI", "ID", "IL", 

  "IN", "IA", "KS", "KY", "LA", "MA", "ME", "MD", "MI", 

  "MN", "MO", "MS", "MT", "NC", "ND", "NE", "NH", "NJ", 

  "NM", "NV", "NY", "OH", "OK", "OR", "PA", "RI", "SC", 

  "SD", "TN", "TX", "UT", "VA", "VT", "WA", "WI", "WY"};

private void Page_Load(object sender, System.EventArgs e)

{

     // Get the connection and adapters from Session state.

     ContactMgmt = (SqlConnection)Session["ContactMgmt"];

     adptContactTypes = (SqlDataAdapter)Session["adptContactTypes"];

     adptContacts = (SqlDataAdapter)Session["adptContacts"];

     // Fill the ContactType data set.

     adptContactTypes.Fill(dsContactTypes, "Contact_Types");

     // Fill the Contacts data set.

     adptContacts.Fill(dsContacts);

     // Bind to data -- populates the drpContactTypes and drpState lists.

     drpContactTypes.DataBind();

     drpStates.DataBind();

}

7. Each contact in the Contacts table is identified by a ContactID. This is the primary key for the table, so it must be unique for each contact. Therefore, before you can add a contact to the Contacts data set, you must obtain a new, unique ContactID from the database. Add the following helper function to the Web form class to get this unique ID before adding the contact:

Visual Basic .NET

' Returns a new, unique ContactID.

Private Function GetNewContactID() As Integer

    Dim cmdNewID As New SqlCommand("SELECT MAX(ContactID)" & _

      " FROM Contacts", ContactMgmt)

    ' Open the database connection.

    ContactMgmt.Open()

    ' Get the highest ID.

    Dim intID As Integer

    ' Make sure records exist.

    If Not (cmdNewID.ExecuteScalar() Is DBNull.Value) Then

    ' Get highest number.

        intID = CInt(cmdNewID.ExecuteScalar())

    Else

        ' Otherwise, set intID to 0.

        intID = 0

    End If

    ' Close the connection

    ContactMgmt.Close()

    ' Return a new ID (one more than highest ID).

    Return intID + 1

End Function

Visual C#

// Returns a new, unique ContactID.

int GetNewContactID()

{

     SqlCommand cmdNewID = new SqlCommand("SELECT " +

     "MAX(ContactID) FROM Contacts", ContactMgmt);

     // Open the database connection.

     ContactMgmt.Open();

     // Get the highest ID.

     int intID;

     // Make sure records exist.

     if (cmdNewID.ExecuteScalar() != DBNull.Value)

         // Get highest number.

         intID = (int)cmdNewID.ExecuteScalar();

     else

        // Otherwise, set intID to 0.

        intID = 0;

     // Close the connection

     ContactMgmt.Close();

     // Return a new ID (one more than highest ID).

     return intID + 1;

}

8. Add the following butAdd_Click event procedure to add the new contact information to the database:

Visual Basic .NET

Private Sub butAdd_Click(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles butAdd.Click

    ' Create a new row for the data set.

    Dim rowNew As dsContacts.ContactsRow

    rowNew = dsContacts.Contacts.NewContactsRow()

    ' Add data to the row.

    rowNew.ContactID = GetNewContactID()

    rowNew.FirstName = txtFirstName.Text

    rowNew.LastName = txtLastName.Text

    rowNew.Address = txtAddress.Text

    rowNew.City = txtCity.Text

    rowNew.StateOrProvince = drpStates.SelectedItem.Text

    rowNew.PostalCode = txtZip.Text

    rowNew.HomePhone = txtHomePhone.Text

    rowNew.WorkPhone = txtWorkPhone.Text

    rowNew.Notes = txtNotes.Text

    rowNew.ContactTypeID = drpContactTypes.SelectedItem.Value

    ' Add the row to the data set.

    dsContacts.Contacts.AddContactsRow(rowNew)

    Try

        ' Modify the database.

        adptContacts.Update(dsContacts)

        ' Show success.

        litStatus.Text = rowNew.FirstName & " " & rowNew.LastName & _

          " added successfully.<br>"

        ' Redisplay page to clear fields.

        Server.Transfer("AddContact.aspx")

    Catch ex As Exception

        litStatus.Text = "The following error occurred:<br>" & _

          ex.Message & "<br>" & _

          "Correct the error and click Add to add the contact " & _

          "or click Cancel to abort.<br>"

    End Try

End Sub

Visual C#

private void butAdd_Click(object sender, System.EventArgs e)

{

     // Create a new row for the data set.

     dsContacts.ContactsRow rowNew;

     rowNew = dsContacts.Contacts.NewContactsRow();

     // Add data to the row.

     rowNew.ContactID = GetNewContactID();

     rowNew.FirstName = txtFirstName.Text;

     rowNew.LastName = txtLastName.Text;

     rowNew.Address = txtAddress.Text;

     rowNew.City = txtCity.Text;

     rowNew.StateOrProvince = drpStates.SelectedItem.Text;

     rowNew.PostalCode = txtZip.Text;

     rowNew.HomePhone = txtHomePhone.Text;

     rowNew.WorkPhone = txtWorkPhone.Text;

     rowNew.Notes = txtNotes.Text;

     rowNew.ContactTypeID = Convert.ToInt16

       (drpContactTypes.SelectedItem.Value);

     // Add the row to the data set.

     dsContacts.Contacts.AddContactsRow(rowNew);

     try

     {

        // Modify the database.

        adptContacts.Update(dsContacts);

        // Show success.

        litStatus.Text = rowNew.FirstName + " " + rowNew.LastName +

          " added successfully.<br>";

        // Redisplay page to clear fields.

        Server.Transfer("AddContact.aspx");

     }

     catch (Exception ex)

     {

        litStatus.Text = "The following error occurred:<br>" +

        ex.Message + "<br>" +

        "Correct the error and click Add to add the contact " +

              "or click Cancel to abort.<br>";

     }

}

9. Add the butCancel_Click event procedure to allow the user to cancel the operation and clear the text boxes on the form:

Visual Basic .NET

Private Sub butCancel_ServerClick(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles butCancel.ServerClick

    ' Redisplay the page to clear text boxes.

     Server.Transfer("AddContact.aspx")

End Sub

Visual C#

private void butCancel_ServerClick(object sender, System.EventArgs e)

{

     // Redisplay the page to clear text boxes.

     Server.Transfer("AddContact.aspx");

}

Exercise 4: Create the Calls Form

The Calls Web form lets users enter and view telephone calls from a contact. It uses a DataList control to view the telephone calls for a contact selected in a DropDownList control, and contains the controls that add a call in a Panel control so they can be hidden and displayed when the user clicks Add.

When completed, the Calls Web form appears as shown in Figure 5-25.

Figure 5-25 The Calls form

To create the Calls Web form

1. Add a new Web form to the project named Calls.aspx.

2. Add controls to the Web form with the property settings shown in the following table. Place the controls shown after the Panel control inside the panel so that you can easily control their display at run time.

	Control
	Property
	Setting

	DropDownList
	ID
	drpContacts

	Button
	ID
	butView

	
	Text
	View Call

	Button
	ID
	butAddCall

	
	Text
	Add Call

	Panel
	ID
	pnlAdd

	TextBox
	ID
	txtDate

	TextBox
	ID
	txtTime

	TextBox
	ID
	txtSubject

	TextBox
	ID
	txtNotes

	
	TextMode
	Multiline

	Button
	ID
	butOKAdd

	Button
	ID
	butCancelAdd

	Literal
	ID
	litError


3. Open the Data tab in the Toolbox and drag a DataSet component onto the Web form. Visual Studio displays the Add Dataset dialog box with the Typed Dataset option selected.

4. In the Name dropdown box, select the typed data set named ContactManagement.dsCalls and click OK. Visual Studio adds the data set to the Web form.

5. From the Web Forms tab in the Toolbox, add a DataList control to the Web form below the panel. Set its properties and edit its templates as shown in the following table:

	Control
	Property
	Setting

	DataList
	ID
	dlstCalls

	
	DataSource
	dsCalls

	DataList Header Template
	
	

	Regular text
	none
	Calls

	HTML Rule
	none
	<HR>

	DataList Footer Template
	
	

	HTML Rule
	none
	<HR>

	DataList Item Template
	
	

	Label
	ID
	lblDate

	
	(DataBindings)
	Using the DataBinding dialog box, choose Container.DataItem.CallDate from the Simple Binding tree. Choose format {0:d}.

	
	
	

	Label
	ID
	lblTime

	
	(DataBindings)
	Using the DataBinding dialog box, choose Container.DataItem.CallTime from the Simple Binding tree. Choose format {0:t}.

	Regular text
	none
	Subject:

	Label
	ID
	lblSubject

	
	(DataBindings)
	Using the DataBinding dialog box, choose Container.DataItem.Subject from the Simple Binding tree.

	Regular text
	none
	Notes:

	Label
	ID
	lblNotes

	
	(DataBindings)
	Using the DataBinding dialog box, choose Container.DataItem.Notes from the Simple Binding tree.

	Separator Template
	
	

	HTML Rule
	none
	<HR>


6. Copy the Hyperlink control from the bottom of the AddContact form to the bottom of the Calls form. To copy the control, select the control and press Ctrl+C. To paste the copy on to the Calls form, click the form in Design mode and press Ctrl+V.

7. Double-click the Calls Web form to display the Code window, and add the following Imports or using statements to the top of the module:

Visual Basic .NET

Imports System.Data.SqlClient

Visual C#

using System.Data.SqlClient;

8. The Calls Web form uses the data components from the Global.asax file, so you need to retrieve references to those components from the state variables created in Exercise 2. The following declarations and Page_Load event procedure retrieve references to the data components and make them available to the form code:

Visual Basic .NET

Dim ContactMgmt As SqlConnection

Dim adptContacts As SqlDataAdapter

Dim adptCalls As SqlDataAdapter

Dim dsContacts As New dsContacts()

Private Sub Page_Load(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles MyBase.Load

    ' Get session state variables.

    ContactMgmt = Session("ContactMgmt")

    adptContacts = Session("adptContacts")

    adptCalls = Session("adptCalls")

    dsCalls = Session("dsCalls")

    dsContacts = Session("dsContacts")

    ' Run the first time page is displayed.

    If Not IsPostBack Then

        ' Fill the Contacts data set.

        adptContacts.Fill(dsContacts)

        ' For each row in the table...

        Dim rowNext As dsContacts.ContactsRow

        For Each rowNext In dsContacts.Contacts

            ' Create a new list item.

            Dim lstNew As New ListItem()

            lstNew.Text = rowNext.FirstName & " " & rowNext.LastName

            lstNew.Value = rowNext.ContactID

            ' Add the list item to the drop-down list.

            drpContacts.Items.Add(lstNew)

        Next

        ' Select the first item in the list.

        drpContacts.SelectedIndex = 0

    End If

End Sub

Visual C#

SqlConnection ContactMgmt;

SqlDataAdapter adptContacts;

SqlDataAdapter adptCalls;

dsContacts dsContacts = new dsContacts();

private void Page_Load(object sender, System.EventArgs e)

{

     // Get session state variables.

     ContactMgmt = (SqlConnection)Session["ContactMgmt"];

     adptContacts = (SqlDataAdapter)Session["adptContacts"];

     adptCalls = (SqlDataAdapter)Session["adptCalls"];

     dsCalls = (dsCalls)Session["dsCalls"];

     dsContacts = (dsContacts)Session["dsContacts"];

     // Run the first time page is displayed.

     if (!IsPostBack)

     {

        // Fill the Contacts data set.

        adptContacts.Fill(dsContacts);

        // For each row in the table...

        foreach (dsContacts.ContactsRow rowNext in dsContacts.Contacts)

        {

              // Create a new list item.

              ListItem lstNew = new ListItem();

              lstNew.Text = rowNext.FirstName + " " + rowNext.LastName;

              lstNew.Value = rowNext.ContactID.ToString();

              // Add the list item to the drop-down list.

              drpContacts.Items.Add(lstNew);

        }

        // Select the first item in the list.

        drpContacts.SelectedIndex = 0;

     }

}

9. The Calls Web form modifies the Calls data set at run time. You need to add the following code to the Page_PreRender event procedure to store the data set back to the state variable after the page is finished processing. (The PreRender event is the last event processed before the page is displayed.)

Visual Basic .NET

Private Sub Page_PreRender(ByVal sender As Object, _

  ByVal e As System.EventArgs) Handles MyBase.PreRender

    ' Save the data set to the session state.

    Session("dsCalls") = dsCalls

End Sub

Visual C#

private void Page_PreRender(object sender, System.EventArgs e)

{

        // Save the data set to the session state.

        Session["dsCalls"] = dsCalls;

}

10. Add the following code for the View and Add button Click event procedures:

Visual Basic .NET

Private Sub butView_Click(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles butView.Click

    ' Set the Select command for the data set.

    adptCalls.SelectCommand.CommandText = "SELECT * FROM Calls " & _

      "WHERE ContactID=" & drpContacts.SelectedItem.Value

    ' Clear the data set.

     dsCalls.Clear

    ' Fill the calls data set.

    adptCalls.Fill(dsCalls)

    ' Bind to the DataList control

    dlstCalls.DataBind()

End Sub

Private Sub butAddCall_Click(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles butAddCall.Click

    ' Set the initial date and time values.

    txtDate.Text = Now.ToShortDateString()

    txtTime.Text = Now.ToShortTimeString()

    ' Make add call panel visible

    pnlAdd.Visible = True

    ' Disable buttons/DropDown list.

    butView.Enabled = False

    butAddCall.Enabled = False

    drpContacts.Enabled = False

End Sub

private void butView_Click(object sender, System.EventArgs e)

{

     // Set the Select command for the data set.

     adptCalls.SelectCommand.CommandText = "SELECT * FROM Calls " +

       "WHERE ContactID=" + drpContacts.SelectedItem.Value;

     // Clear the data set.

     dsCalls.Clear();

     // Fill the calls data set.

     adptCalls.Fill(dsCalls);

     // Bind to the DataList control

     dlstCalls.DataBind();

}

private void butAddCall_Click(object sender, System.EventArgs e)

{

     // Set the initial date and time values.

     txtDate.Text = System.DateTime.Now.ToShortDateString();

     txtTime.Text = System.DateTime.Now.ToShortTimeString();

     // Make add call panel visible

     pnlAdd.Visible = true;

     // Disable buttons/DropDown list.

     butView.Enabled = false;

     butAddCall.Enabled = false;

     drpContacts.Enabled = false;

}

11. Add the following helper function to get a new CallID value. Because CallID is the primary key in the Calls table, you need a value that is unique within the database, as shown here:

' Returns a new, unique CallID.

Private Function GetCallID() As Integer

    Dim cmdNewID As New SqlCommand("SELECT MAX(CallID)" & _

      " FROM Calls", ContactMgmt)

    ' Open the database connection.

    ContactMgmt.Open()

    ' Get the highest ID.

    Dim intID As Integer

    ' Make sure records exist.

    If Not (cmdNewID.ExecuteScalar() Is DBNull.Value) Then

        ' Get highest number.

        intID = CInt(cmdNewID.ExecuteScalar())

    Else

        ' Otherwise, set intID to 0.

        intID = 0

    End If

    ' Close the connection

    ContactMgmt.Close()

    ' Return a new ID (one more than highest ID).

    Return intID + 1

End Sub

int GetCallID()

{

     SqlCommand cmdNewID = new SqlCommand("SELECT MAX(CallID)" +

       " FROM Calls", ContactMgmt);

     // Open the database connection.

     ContactMgmt.Open();

     // Get the highest ID.

     int intID;

     // Make sure records exist.

     if (cmdNewID.ExecuteScalar() != DBNull.Value)

        // Get highest number.

        intID = (int)cmdNewID.ExecuteScalar();

     else

        // Otherwise, set intID to 0.

        intID = 0;

     // Close the connection

     ContactMgmt.Close();

     // Return a new ID (one more than highest ID).

     return intID + 1;

}

12. Add the following code to the OK and Cancel button event procedures. These event procedures are very similar to the ones used to add or cancel the adding of records in the Contacts table in the previous exercise.

Private Sub butOKAdd_Click(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles butOKAdd.Click

    Dim rowNew As dsCalls.CallsRow

    rowNew = dsCalls.Calls.NewCallsRow

    rowNew.ContactID = drpContacts.SelectedItem.Value

    rowNew.CallID = GetCallID()

    rowNew.CallDate = txtDate.Text

    rowNew.CallTime = txtDate.Text & " " & txtTime.Text

    rowNew.Subject = txtSubject.Text

    rowNew.Notes = txtNotes.Text

    ' Add row to calls data set.

    dsCalls.Calls.AddCallsRow(rowNew)

    Try

        adptCalls.Update(dsCalls)

        ' Hide the add call panel.

        pnlAdd.Visible = False

        ' Clear the fields.

        txtSubject.Text = ""

        txtNotes.Text = ""

        ' Enable the other controls.

        butView.Enabled = True

        butAddCall.Enabled = True

        drpContacts.Enabled = True

    Catch ex As Exception

        ' Display error.

        litError.Text = "The following error occurred while adding " & _

            "the call:<br>" & _

            ex.Message & "<br>" & _

            "Correct the error and try again, or click Cancel to " & _

            "abort.<br>"

    End Try

End Sub

Private Sub butCancelAdd_Click(ByVal sender As System.Object, _

  ByVal e As System.EventArgs) Handles butCancelAdd.Click

    ' Hide the add call panel.

    pnlAdd.Visible = False

    ' Clear the fields.

    txtSubject.Text = ""

    txtNotes.Text = ""

    ' Enable the other controls.

    butAddCall.Enabled = True

    drpContacts.Enabled = True

End Sub

private void butOKAdd_Click(object sender, System.EventArgs e)

{

     dsCalls.CallsRow rowNew;

     rowNew = dsCalls.Calls.NewCallsRow();

     rowNew.ContactID = Convert.ToInt16(drpContacts.SelectedItem.Value);

     rowNew.CallID = GetCallID();

     rowNew.CallDate = Convert.ToDateTime(txtDate.Text);

     rowNew.CallTime = Convert.ToDateTime(txtDate.Text + " "

        + txtTime.Text);

     rowNew.Subject = txtSubject.Text;

     rowNew.Notes = txtNotes.Text;

     // Add row to calls data set.

     dsCalls.Calls.AddCallsRow(rowNew);

     try

     {

        adptCalls.Update(dsCalls);

        // Hide the add call panel.

        pnlAdd.Visible = false;

    // Clear the fields.

    txtSubject.Text = "";

    txtNotes.Text = "";

        // Enable the other controls.

        butView.Enabled = true;

        butAddCall.Enabled = true;

        drpContacts.Enabled = true;

     }

     catch (Exception ex) 

     {

        // Display error.

        litError.Text = "The following error occurred while adding " +

        "the call:<br>" +

        ex.Message + "<br>" +

        "Correct the error and try again, or click Cancel to " +

        "abort.<br>";

     }

}

private void butCancelAdd_Click(object sender, System.EventArgs e)

{

     // Hide the add call panel.

     pnlAdd.Visible = false;

    // Clear the fields.

    txtSubject.Text = "";

    txtNotes.Text = "";

     // Enable the other controls.

     butAddCall.Enabled = true;

     drpContacts.Enabled = true;

}

Exercise 5: Create the DeleteContact and ContactTypes Forms

So far this lab has shown you how to create Web forms in a step-by-step fashion. Now it' s time to strike out on your own! Create the DeleteContact and ContactTypes Web forms by yourself. These Web forms should perform the following tasks:

· Use the data components from the Global.asax file.

· Perform operations on the database. The DeleteContact form should allow users to delete contacts from the database. The ContactTypes form should allow users to view and add new contact types.

· Use data set transactions, as demonstrated by the AddContact and Calls forms.

· Provide navigation back to the SwitchBoard form.

